スロットルモデルの特性理解(1)
～スロットルバルブ開口面積～

本ワークシートでの内容

- スロットルバルブ開口面積モデルで利用されている関数のグラフを表示したり、パラメータを変更してグラフを表示することで、モデルの特性を知ることができます

目次

スロットルバルブ開口面積モデル

ここでは、スロットルバルブのモデリングを行うことを考えます。

モデルの概要

まず、スロットル断面積を以下の式で定義します。

スロットルバルブの開口面積



図１：スロットバルブ

スロットルバルブの開口面積は、以下の式で定義されます。(参考文献 [1])

- $\phi < \arccos(a\cos(\phi 0))$の場合

$$\mathrm{Athr} = -\frac{d\,Dt}{2}\left(1-a^2\right)^{\frac{1}{2}} + \frac{d\,Dt}{2}\sqrt{1-\left(\frac{a\cos(\phi 0)}{\cos(\phi)}\right)^2} + \frac{Dt^2}{2}\arcsin\left(\left(1-a^2\right)^{\frac{1}{2}}\right) - \frac{Dt^2}{2}\,\frac{\cos(\phi)\arcsin\left(1-\left(\frac{a\cos(\phi 0)}{\cos(\phi)}\right)^2\right)}{\cos(\phi 0)}$$

- その他（すなわち $\phi \geq \arccos(a\cos(\phi 0))$ の場合）

$$\mathrm{Athr} = \frac{1}{2}Dt^2\arcsin\left(\left(1-a^2\right)^{\frac{1}{2}}\right) - \frac{1}{2}d\,Dt\left(\left(1-a^2\right)^{\frac{1}{2}}\right)$$

ただし、各変数は以下のとおりです。

d= スロットルバルブピンの直径（m）
Dt= 絞り板の直径 (m)
ϕ_0 = スロットバルブが閉じる角度 (degree)　（内部にて rad に変換して利用）
ϕ= スロットルバルブ角度 (degree)
A_{thr} = スロットル開口面積 (m)

また、変数 a は

$$a = \frac{d}{Dt}$$

とします。

• スロットルバルブの開口面積グラフ表示（アプリケーション作成例）

以下のようにGUIアプリケーションを作成することで、値を入力したり、スライダーで値を変更して、簡単にグラフの特徴をつかむことができます。次節以降で、モデルの設定や下記のグラフを表示して関数の特徴を調査する方法を紹介します。

スロットルバルブの開口面積のグラフ



初期設定に戻す

変数名	スライダー設定	値	説明
d	0.0　0.005　0.01	0. 005000	スロットルバルブピンの直径（m） 初期値：0.005 (m)
Dt	0.05 0.06 0.07 0.08 0.09 0.1	0. 075000	絞り板の直径 (m) 初期値: 0.075 (m)
θ_0	0　5　10　15　20	8	スロットバルブが閉じる角度 (°) ($\phi_0 = \pi * \theta_0 / 180$ にて内部では変換) 初期値：8 (°)

モデルの設定

以下、モデルを設定します。なお、入力間違いを避けるため、以下 1DMath での入力で説明を行います。

• 区分関数部分の設定
ここでは、以下の内容に分けて入力を行います。

【条件式の設定】

```
> cd1:= phi < arccos(a*cos(phi0));
```

$$cd1 := \phi < \arccos(a \cos(\phi 0)) \tag{1}$$

【条件式に対応する式の設定】

• 第1項

```
> f1a := -(1/2)*d*Dt*(1-a^2)^(1/2);
```

$$f1a := -\frac{1}{2}\, d\, Dt \sqrt{-a^2+1} \tag{2}$$

• 第2項

```
> f1b := (1/2)*d*Dt*(1-(a*cos(phi0)/cos(phi))^2)^(1/2);
```

$$f1b := \frac{1}{2}\, d\, Dt \sqrt{1-\frac{a^2 \cos(\phi 0)^2}{\cos(\phi)^2}} \tag{3}$$

• 第3項

```
> f1c := (1/2)*Dt^2*arcsin((1-a^2)^(1/2));
```

$$f1c := \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) \tag{4}$$

• 第4項（負の符号も考慮）

```
> f1d := -(1/2)*Dt^2*cos(phi)*arcsin((1-(a*cos(phi0)/cos(phi))^2)^
  (1/2))/cos(phi0);
```

$$f1d := -\frac{1}{2}\, \frac{Dt^2 \cos(\phi) \arcsin\left(\sqrt{1-\frac{a^2 \cos(\phi 0)^2}{\cos(\phi)^2}}\right)}{\cos(\phi 0)} \tag{5}$$

• 第1項から第4項までを1つの式にまとめます。

```
> f1 := f1a + f1b + f1c + f1d;
```

$$f1 := -\frac{1}{2}\, d\, Dt \sqrt{-a^2+1} + \frac{1}{2}\, d\, Dt \sqrt{1-\frac{a^2 \cos(\phi 0)^2}{\cos(\phi)^2}} + \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) - \frac{1}{2}\, \frac{Dt^2 \cos(\phi) \arcsin\left(\sqrt{1-\frac{a^2 \cos(\phi 0)^2}{\cos(\phi)^2}}\right)}{\cos(\phi 0)} \tag{6}$$

【デフォルト時（その他の場合）の式の設定】

• 第1項

```
> f2a := (1/2)*Dt^2*arcsin((1-a^2)^(1/2));
```

$$f2a := \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) \tag{7}$$

• 第2項（負の符号も考慮）

```
> f2b := -(1/2)*d*Dt*(1-a^2)^(1/2);
```

$$f2b := -\frac{1}{2}\, d\, Dt\, \sqrt{-a^2+1} \qquad \textbf{(8)}$$

- 上の2つの項を1つの式にまとめます。

```
> f2 := f2a + f2b;
```

$$f2 := \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) - \frac{1}{2}\, d\, Dt\, \sqrt{-a^2+1} \qquad \textbf{(9)}$$

【区分関数の設定】

piecewiseコマンドを用いて区分関数を生成します。

```
> tfunc := Athr=piecewise(cd1, f1, f2);
```

$$tfunc := Athr = \begin{cases} -\frac{1}{2}\, d\, Dt\, \sqrt{-a^2+1} + \frac{1}{2}\, d\, Dt\, \sqrt{1-\frac{a^2\cos(\phi 0)^2}{\cos(\phi)^2}} + \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) - \frac{1}{2} \\ \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) - \frac{1}{2}\, d\, Dt\, \sqrt{-a^2+} \end{cases}$$

- モデルの設定

上記の区分関数と合わせて、スロットルバルブの開口面積の式を定義します。

```
> eq := [a=d/Dt, tfunc];
```

$$eq := \left[a = \frac{d}{Dt},\ Athr = \left\{ \left[\left[-\frac{1}{2}\, d\, Dt\, \sqrt{-a^2+1} + \frac{1}{2}\, d\, Dt\, \sqrt{1-\frac{a^2\cos(\phi 0)^2}{\cos(\phi)^2}} + \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) - \frac{1}{2}\, \frac{Dt^2 \cos(\phi) \arcsin\left(\sqrt{1-\frac{a^2\cos(\phi 0)^2}{\cos(\phi)^2}}\right)}{\cos(\phi 0)},\ \phi < \arccos(a\cos(\phi 0)) \right], \left[\frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-a^2+1}\right) - \frac{1}{2}\, d\, Dt\, \sqrt{-a^2+1},\ otherwise \right] \right] \right\} \right] \qquad \textbf{(11)}$$

- パラメータの設定

利用するパラメータ
d=0.005 (m) D=0.075 (m) $\phi_0 = 8$ (degree) $= \frac{2}{45}\pi$ (radian)

ここでは上記のパラメータを利用します。

```
> params := [d=0.005, Dt=0.75e-1, phi0=convert(8*degrees, radians)]
  ;
```

$$params := \left[d=0.005,\ Dt=0.075,\ \phi 0=\frac{2}{45}\,\pi\right] \tag{12}$$

スロットルバルブ開口面積モデルの特性理解

設定された式のグラフを表示したり、パラメータを変更して表示することで関数の特徴をつかむことができます。

グラフ表示

$a=\frac{\mathrm{d}}{\mathrm{Dt}}$ および パラメータの値を代入すると、以下の関数を得ます。

方程式の各式を確認します。

```
> eq;
```

$$\left[a=\frac{d}{Dt},\ Athr=\left\{\begin{array}{ll} -\frac{1}{2}\,d\,Dt\,\sqrt{-a^2+1}+\frac{1}{2}\,d\,Dt\,\sqrt{1-\frac{a^2\cos(\phi 0)^2}{\cos(\phi)^2}}+\frac{1}{2}\,Dt^2\arcsin\left(\sqrt{-a^2+1}\right)-\frac{1}{2}\,\frac{Dt^2\cos(\phi)\,\arcsin\left(\sqrt{1-\frac{a^2\cos(\phi 0)^2}{\cos(\phi)^2}}\right)}{\cos(\phi 0)}, & \phi<\arccos(a\cos(\phi 0)) \\ \frac{1}{2}\,Dt^2\arcsin\left(\sqrt{-a^2+1}\right)-\frac{1}{2}\,d\,Dt\,\sqrt{-a^2+1}, & otherwise\end{array}\right.\right] \tag{13}$$

第1式 を 第2式 に代入した式を、変数に保存します。（変数 a について削除した式を表します。**【この式は後で次節でも利用します】**）

```
> eq[1];
```

$$a=\frac{d}{Dt} \tag{14}$$

```
> eq[2];
```

$$Athr=\left\{\begin{array}{ll} -\frac{1}{2}\,d\,Dt\,\sqrt{-a^2+1}+\frac{1}{2}\,d\,Dt\,\sqrt{1-\frac{a^2\cos(\phi 0)^2}{\cos(\phi)^2}}+\frac{1}{2}\,Dt^2\arcsin\left(\sqrt{-a^2+1}\right)-\frac{1}{2}\,\frac{Dt^2\cos}{} \\ \frac{1}{2}\,Dt^2\arcsin\left(\sqrt{-a^2+1}\right)-\frac{1}{2}\,d\,Dt\,\sqrt{-a^2+1}\end{array}\right.$$

```
> tmpf := subs(eq[1], eq[2]);
```

$$tmpf := Athr = \begin{cases} -\frac{1}{2}\, d\, Dt \sqrt{-\frac{d^2}{Dt^2}+1} + \frac{1}{2}\, d\, Dt \sqrt{1 - \frac{d^2 \cos(\phi 0)^2}{Dt^2 \cos(\phi)^2}} + \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-\frac{d^2}{Dt^2}+1 \\ \frac{1}{2}\, Dt^2 \arcsin\left(\sqrt{-\frac{d^2}{Dt^2}+1}\right) - \frac{1}{2}\, d\, Dt \sqrt{- \end{cases}$$

パラメータ値を代入する。

```
> func1 := subs(params,tmpf);
```

$$func1 := Athr = \begin{cases} -0.0001870828694 + 0.0001875000000 \sqrt{1 - \frac{0.004444444444 \cos\left(\frac{2}{45}\pi\right)^2}{\cos(\phi)^2}} + 0.00 \\ 0.002812500000 \end{cases}$$

- 角度 ϕ をラジアンから度（°）に置き換えて表示

右辺の式を取り出して、$\phi = \frac{\pi \cdot \theta}{180}$ として横軸方向の角度を度（°）の変数に置き換えます。

```
> tmp2 := subs(phi=Pi*theta/180, func1):
> func1B := rhs(tmp2);
```

$$func1B := \begin{cases} 0.004043142634 + 0.0001875000000 \sqrt{1 - \frac{0.004444444444 \cos\left(\frac{2}{45}\pi\right)^2}{\cos\left(\frac{1}{180}\pi\,\theta\right)^2}} - \frac{0.00281250}{} \\ 0.004043142634 \end{cases}$$

- plotコマンドで表示

上記で求めた関数のグラフを表示します。

入力例

```
> plot(func1B, theta=0..90, view = 0..0.0045);
```



入力例（応用）

以下、表示例を紹介します。上記のplotコマンドでの表示に、以下のオプションを追加します。

- labels: 両軸のラベル
- labeldirections: 両軸のラベルの位置
- labelfont: 両軸のラベルのフォント
- size: プロットウィンドウのサイズ

• オプション設定を追加したグラフ表示

```
> plot(func1B, theta=0..90, view = 0..0.0045,
  labels = ["Throttle Angle (Degree)", "Throttle Area(m^2)"],
  labeldirections = ["horizontal", "vertical"],
  labelfont = ["HELVETICA", 10],
  size=[300, 300]);
```



Exploreコマンドを利用した関数の調査

スロットルの絞り板の直径 Dt の値を変化させてスロットル開口面積がどのように変化するかを調べます。ここでは、Exploreコマンドを活用します。

- パラメータを設定（今回は Dt の値を除きます）

```
> params2:=[d=0.5e-2, phi0=convert(8*degrees, radians)];
```

$$params2 := \left[d = 0.005,\ \phi 0 = \frac{2}{45}\pi\right] \tag{19}$$

- 設定したモデルから 変数 a を消去した式の右辺を表示（変数 tmpf は上記で計算済みです）

```
> rhs(tmpf);
```

$$-\frac{1}{2} d\, Dt \sqrt{-\frac{d^2}{Dt^2}+1} + \frac{1}{2} d\, Dt \sqrt{1-\frac{d^2 \cos(\phi 0)^2}{Dt^2 \cos(\phi)^2}} + \frac{1}{2} Dt^2 \arcsin\left(\sqrt{-\frac{d^2}{Dt^2}+1}\right) - \frac{1}{2} \frac{Dt^2\ \mathrm{co}}{}$$

$$\frac{1}{2} Dt^2 \arcsin\left(\sqrt{-\frac{d^2}{Dt^2}+1}\right) - \frac{1}{2} d\, Dt \sqrt{-\frac{d^2}{Dt^2}+1}$$

- パラメータを代入

上式にパラメータを代入します。

```
> tmp3 := subs(params2, rhs(tmpf));
```

$$tmp3 := \left\{ \begin{array}{l} -0.002500000000\, Dt \sqrt{-\frac{0.000025}{Dt^2}+1} + 0.002500000000\, Dt \sqrt{1 - \frac{0.000025 \cos\left(\frac{2}{45}\pi\right)^{2}}{Dt^2 \cos(\phi)^2}} \\ \frac{1}{2} Dt^2 \arcsin\left(\sqrt{-\frac{0.000025}{Dt^2}+1}\right) \end{array} \right.$$

• 関数の作成

"->" （アロー演算子）利用して、変数 ϕ と Dt を引数とする関数を作成します。

```
> func2B:= (Dt, phi)->(21);
```

$$func2B := (Dt, \phi) \rightarrow piecewise\left(\phi < \arccos\left(\frac{0.005 \cos\left(\frac{2}{45}\pi\right)}{Dt}\right), -0.002500000000\, Dt \sqrt{-\frac{0.000025}{Dt^2}+1} + 0.002500000000\, Dt \sqrt{1 - \frac{0.000025 \cos\left(\frac{2}{45}\pi\right)^2}{Dt^2 \cos(\phi)^2}} + \frac{1}{2} Dt^2 \arcsin\left(\sqrt{-\frac{0.000025}{Dt^2}+1}\right) - \frac{1}{2} \frac{Dt^2 \cos(\phi) \arcsin\left(\sqrt{1 - \frac{0.000025 \cos\left(\frac{2}{45}\pi\right)^2}{Dt^2 \cos(\phi)^2}}\right)}{\cos\left(\frac{2}{45}\pi\right)}, \frac{1}{2} Dt^2 \arcsin\left(\sqrt{-\frac{0.000025}{Dt^2}+1}\right) - 0.002500000000\, Dt \sqrt{-\frac{0.000025}{Dt^2}+1}\right) \quad (22)$$

• Exploreコマンドの利用

Exploreコマンドを利用して、変数 Dt の値をスライダーで変更できるように設定します。

```
> Explore(plot(func2B(Dt, Pi*theta/180), theta=0..90, view=0.
  .0.005),
  parameters=[Dt=0.04..0.08],
  initialvalues=[Dt=0.75e-1],
  size=[250, 250]);
```



- アプリケーション化

GUIコンポーネントを利用して上記のグラフをアプリケーションを作成することができます。作成例は、このワークシートの最初で紹介しているのでそちらを参照してください。

主な利用コマンド

コマンド名/演算子名	説明
• piecewise(条件式1, 式1, 条件式2, 式2, .. ., デフォルトの数式)	区分関数の定義 使用例: piecewise$(x < -1$ and $1 < x, 1, 0)$
• subs(部分式, 式)	部分式の代入 使用例: subs$(s = x^2, s + 1)$
• convert(度*degrees, radians)	角度の変換（度（°）からラジアンへの変換）

	使用例: `convert(90*degrees, radians)`
• **plot(関数, 横軸範囲, オプション)** • **plot([関数1, 関数2, ...], 横軸範囲, オプション)**	2次元グラフを作成 複数の関数の2次元グラフを作成 <plotオプション> - labels: 両軸のラベル - labeldirections: 両軸のラベルの位置 - labelfont: 両軸のラベルのフォント - size: プロットウィンドウのサイズ *その他のオプションに関してはMapleヘルプ「plot, options」で検索/参照
• **Explore(式, オプション)**	パラメータに依存する数式に対して式の特徴を調査

参考文献

1. Moskwa, J.J, Automotive Engine Modeling for real Time Control, Ph.D. Dissertation, Massachusetts Institute of Technology, 1988.
2. Guzzella, L, Onder, C, Introduction to Modeling and Control of Internal Combustion Engine Systems, Springer, 2009.
3. 申 鉄龍、大畠 明 編著, 自動車エンジンのモデリングと制御、コロナ社, 2011.